§3.10-3.12: A fun example: ROT13

devo

29 Sep 2005
CMPT14x
Dr. Sean Ho
Trinity Western University

Reminders:

1) journals in folder

Review of 4.1-4.2

- Helper functions: ABS, CAP, INC, DEC
- Qualified IMPORT
- Procedures:
  - No parameters
  - Read-only parameters
  - Writeable parameters
  - Both kinds of parameters
  - Formal vs. actual parameters
  - Scope

A fun example: ROT13

- Task: Translate characters into ROT13 one line at a time
  - ROT13:
    - Treat each letter A-Z as a number between 1-26,
    - Add 13 to the number and wrap-around if necessary
    - Convert back to a letter
    - Preserve case
    - Leave all non-letter characters alone
  - e.g., ROT13 ('a') = 'n', ROT13 ('P') = 'C', ROT13 ('#') = '#'

ROT13: Problem restatement

- Input:
  - A sequence of letters, ending with a newline
- Computation:
  - Convert letter to numerical form
  - Add 13 and wrap-around if necessary
  - Convert back to letter form
- Output:
  - Print ROT13'd character to screen

ROT13: convert letters to numbers

- How do we convert from a letter character to a numerical code?
  - Try VAL (CARDINAL, char1): testbed program

```
ReadChar (char1);
WriteString (“The numerical ASCII code for ”);
WriteChar (char1);
WriteString (“ is”);
WriteCard (VAL (CARDINAL, char1), 0);
WriteLn;
```

- ASCII codes: 'A' = 65, 'Z' = 90, 'a' = 97, 'z' = 122
- Convert back with VAL (CHAR, card1)

ROT13: Pseudocode

- Print intro to the user
- Repeat:
  - Read a character the user typed
  - Convert to ASCII numerical code
  - If character is an uppercase letter,
    - Add 13 to code
    - If code is now beyond 'Z', subtract 26 (wrap-around)
  - Else if character is a lowercase letter,
    - Add 13 to code
    - If code is now beyond 'z', subtract 26 (wrap-around)
  - Convert numerical code back to character and print
- Until current character is newline

How to test if upper/lower case?

- Our pseudocode involves a test if the character is an uppercase letter or lowercase letter
- How to do that?

```
ascii := VAL (CARDINAL, ch);
IF (ascii >= VAL (CARDINAL, 'A')) AND
    (ascii <= VAL (CARDINAL, 'Z'))
    THEN
        (* uppercase *)
```

ROT13: Stub program pseudocode

- Repeat:
  - Read a character the user typed
  - Convert to ASCII numerical code
  - Convert back to character
  - Print ASCII code and converted character
- Until current character is newline

- This stub program allows us to test the char<->ASCII conversion process and the interactive keyboard reading
- Tackle the ROT13 processing later

ROT13: Stub program code

```
MODULE Rot13Stub;

FROM STextIO IMPORT
    ReadChar, WriteChar;
FROM SWholeIO IMPORT
    WriteCard;
FROM SIOResult IMPORT          (* needed to test for newline *)
    ReadResult, ReadResults;

VAR
    ch : CHAR;                 (* user input *)
    ascii : CARDINAL;          (* numerical code corresponding to ch *)
```

ROT13: Stub program code, p.2

```
BEGIN
    ReadChar (ch);
    WHILE ReadResult() <> endOfLine (* Read until end of line *)
        DO
            ascii := VAL (CARDINAL, ch);
            WriteCard (ascii, 0);
            WriteChar (VAL (CHAR, ascii));
            ReadChar (ch);
        END;
END Rot13Stub.
```

- Sample input: hiya<newline>
- Sample output: “ 104h 105i 121y 97a”

ROT13: Full program code

```
MODULE Rot13;
FROM STextIO IMPORT
    ReadChar, WriteChar, SkipLine;
FROM SIOResult IMPORT
    ReadResult, ReadResults;
CONST
    ASCIIA = VAL (CARDINAL, 'A');      (* ASCII code for 'A' *)
    ASCIIZ = VAL (CARDINAL, 'Z');
    ASCIIa = VAL (CARDINAL, 'a');
    ASCIIz = VAL (CARDINAL, 'z');
VAR
    ch : CHAR;                  (* user input *)
    ascii : CARDINAL;       (* numerical code corresponding to ch *)
```

ROT13: Full program code, p.2

```
BEGIN
    ReadChar (ch);
    WHILE ReadResult() <> endOfLine      (* Read until end of line *)
        DO
            ascii := VAL (CARDINAL, ch);
            IF (ascii >= ASCIIA) AND (ascii <= ASCIIZ) (* uppercase *)
                THEN
                    ascii := ascii + 13;
                    IF (ascii > ASCIIZ)                    (* wrap-around *)
                        THEN
                            ascii := ascii - 26;
                        END;
```

ROT13: Full program code, p.3

```
            ELSIF (ascii >= ASCIIa) AND (ascii <= ASCIIz) THEN
                ascii := ascii + 13;                    (* lowercase *)
                IF (ascii > ASCIIz)                     (* wrap-around *)
                    THEN
                        ascii := ascii - 26;
                    END;
                END;

            WriteChar (VAL (CHAR, ascii));
            ReadChar (ch);
        END;
END Rot13.
```

ROT13: Results and analysis

- Input: hiya
  - Output: uvln
- Input: uvln
  - Output: hiya
- Input: Hello World! This is a longer example.
  - Output: Uryyb Jbeyq! Guvf vf n ybatre rknzcyr.

- Generalizations/extensions?
  - Handle multiple lines one line at a time?
    - How to quit?

Review of today (3.4-3.8)

- ROT13 example:
  - Stub program
  - Using ReadResult() to test the previous Read operation

TODO items

- Homework: §3.14 #17, 36 (hand in on Fri)
- Reading: through §4.7 for Fri